

Description

Here we demonstrate how to obtain goodness-of-fit statistics for latent class models.

We continue with [SEM] [Example 50g](#), where we fit a two-class model:

```
. use https://www.stata-press.com/data/r19/gsem_lca1
. gsem (accident play insurance stock <- ), logit lclass(C 2)
```

See [Latent class models](#) in [SEM] [Intro 5](#) for background.

Remarks and examples

Remarks are presented under the following headings:

[Likelihood-ratio \(\$G^2\$ \) test](#)
[Comparing models](#)

Likelihood-ratio (G^2) test

For standard latent class models with observed variables that are all categorical, one way to evaluate model fit is to compare the model we have just fit with a saturated model. We can use the `estat lcgof` command to perform a likelihood-ratio test of whether our model fits as well as the saturated model. The corresponding likelihood-ratio statistic is sometimes referred to as G^2 in latent class analysis literature.

```
. estat lcgof
```

Fit statistic	Value	Description
Likelihood ratio		
chi2_ms(6)	2.720	model vs. saturated
p > chi2	0.843	
Information criteria		
AIC	1026.935	Akaike's information criterion
BIC	1057.313	Bayesian information criterion

We fail to reject the null hypothesis that our model fits as well as the saturated model.

`estat lcgof` also reports Akaike's information criterion (AIC) and Schwarz's Bayesian information criterion (BIC). These are useful for comparing models but not useful for determining goodness-of-fit for a single model.

Comparing models

In latent class analysis, we often compare models that have different numbers of classes. Following [Goodman \(2002\)](#), we compare models that allow for one, two, and three latent classes. We have already fit the two-class model using the `gsem` command above. Before we move on, we will store the results of this model.

```
. estimates store twoclass
```

Next, we fit the one-class model, store the results, and perform the likelihood-ratio test comparing it with the saturated model.

```
. quietly gsem (accident play insurance stock <- ), logit lclass(C 1)
. estimates store oneclass
. estat lcgof
```

Fit statistic	Value	Description
Likelihood ratio		
chi2_ms(11)	81.084	model vs. saturated
p > chi2	0.000	
Information criteria		
AIC	1095.300	Akaike's information criterion
BIC	1108.801	Bayesian information criterion

We reject the null hypothesis in this case. The one-class model does not fit well.

We also fit the three-class model.

```
. quietly gsem (accident play insurance stock <- ), logit lclass(C 3)
. estimates store threeclass
. estat lcgof
```

Fit statistic	Value	Description
Likelihood ratio		
chi2_ms(1)	0.387	model vs. saturated
p > chi2	0.534	
Information criteria		
AIC	1034.602	Akaike's information criterion
BIC	1081.856	Bayesian information criterion

Based on this test, the three-class model, like the two-class model, does not fit worse than the saturated model.

Now we use `lcstats` to more directly compare our three models.

```
. lcstats oneclass twoclass threeclass
Latent class statistics
```

	Classes	N	ll	Rank	Entropy	df	LMR	P>LMR
oneclass	1	216	-543.65	4				
twoclass	2	216	-504.47	9	0.7193	5	75.55	<0.001
threeclass	3	216	-503.30	14	0.6110	5	2.25	0.687

LMR is the Lo-Mendell-Rubin-adjusted likelihood-ratio test statistic. Likelihood-ratio tests compare the given model versus the same model with one less latent class.

lcstats reports the sample size, log likelihood, and rank for each fitted model. It also reports entropy, a measurement of class separation, for models with 2 or more latent classes. Larger entropy values, closer to 1, correspond to better separation of classes. The specified estimates differ only in the number of latent classes, each having one more latent class than the previous, so lcstats also reports the Lo–Mendell–Rubin (LMR)-adjusted likelihood-ratio test for two scenarios.

1. The first is reported in the row labeled `twoclass`, comparing this model with two latent classes versus `oneclass` with one latent class. We find evidence that the two-class model fits better than the one-class model.
2. The second scenario is reported in the row labeled `threeclass`, comparing this model with three latent classes versus `twoclass` with two latent classes. We do not find evidence that the three-class model fits better than the two-class model.

lcstats has options for reporting the usual information criteria. Here we add the `aic` and `bic` options to get Akaike’s information criterion (AIC) and the Bayesian information criterion (BIC). Adding these statistics makes the table wide, so we also add the `split` option to request that lcstats partition the reported statistics into two tables.

```
. lcstats oneclass twoclass threeclass, aic bic split
Latent class statistics
```

	N	Rank	AIC	BIC	Entropy
oneclass	216	4	1,095.30	1,108.80	
twoclass	216	9	1,026.94	1,057.31	0.7193
threeclass	216	14	1,034.60	1,081.86	0.6110

AIC is the Akaike information criterion.
BIC is the Bayesian information criterion.
BIC uses N = number of observations.

	Classes	ll	df	LMR	P>LMR
oneclass	1	-543.65			
twoclass	2	-504.47	5	75.55	<0.001
threeclass	3	-503.30	5	2.25	0.687

LMR is the Lo–Mendell–Rubin-adjusted likelihood-ratio test statistic.
Likelihood-ratio tests compare the given model versus the same model with one less latent class.

Smaller values of AIC and BIC are better. The two-class model has both the smallest AIC and the smallest BIC.

Reference

Goodman, L. A. 2002. “Latent class analysis: The empirical study of latent types, latent variables, and latent structures”. In *Applied Latent Class Analysis*, edited by J. A. Hagenaars and A. L. McCutcheon, 3–55. Cambridge: Cambridge University Press. <https://doi.org/10.1017/CBO9780511499531.002>.

Also see

[SEM] [Example 50g](#) — Latent class model

[SEM] [Intro 5](#) — Tour of models

[SEM] [gsem](#) — Generalized structural equation model estimation command

[SEM] [estat lcgof](#) — Latent class goodness-of-fit statistics

[SEM] [lcstats](#) — Latent class model-comparison statistics

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

